Generated by: calypso_user, DataServer: localhost Time: Tue Jul 09 07:37:34 EDT 2024
Name: EquitySingleReverseConvertibleSwapv1
Calculator: ForwardPayoff
Description: Reverse Convertible on single underlying with KO

Script: Variables

```
Constant Start As ReferenceDate From Product.StartDate
Constant CouponPeriod As AccrualPeriod[]
Constant Maturity As PaymentDate From Product.Maturity
Constant Settlement As Enum 'Physical', 'Cash'
Constant KI As ReferenceDate[]
Constant KO As AccrualPeriod[]
Constant PrincipalProtection As Double From 1
Constant ITMParticipation As Double From 1
Constant StrikePct As Double From 1
Constant KI_BarrierPct As Double From 0.5
Constant KO_BarrierPct As Double From 1.5
Constant KO_Curr As Currency
Constant KO_FX As Double From 1
Constant KO_Rebate As Double From 0.1
Constant CouponRateStrikePct As Double From 1
Constant FinalCouponRate As Double From 0.0
Constant CouponRateHigh As Double From 0.1
Constant CouponRateLow As Double From 0
Constant PayRec As Integer From Product.BuySell
Constant Curr As Currency From Product.Currency
Constant Notional As Double From Product.Notional
Constant Index As Quotable From Product.Underlying
AboveKO  As Boolean
AboveCP  As Boolean
DelivQty  As Double
RealizedCoupon  As Double
FlowValue  As Double
KNOCKED_IN  As Boolean
KNOCKED_OUT  As Boolean
Strike  As Double
CouponRateStrike As Double
KI_Barrier  As Double
KO_Barrier  As Double
Performance  As Double
Prob_KI As Measure
Prob_KO As Measure
Option As Measure to NPV
Constant IR_CouponPeriod As AccrualPeriod[]
Constant IR_CouponRate As Quotable
Constant IR_Spread As Double
EQLeg As Measure
IRLeg As Measure
```

Script: Forward

```
Start:
  Strike = (StrikePct * Index)
  KO_Barrier = (KO_BarrierPct * Index)
  KI_Barrier = (KI_BarrierPct * Index)
  CouponRateStrike = (CouponRateStrikePct * Index)
KI:
  If Not(KNOCKED_IN) Then
    If (Index <= KI_Barrier) Then
```

```
      KNOCKED_IN = True
    EndIf
  EndIf
CouponPeriod:
  If Not(KNOCKED_OUT) Then
    AboveCP = True
    If (Index < CouponRateStrike) Then
      AboveCP = False
    EndIf
    RealizedCoupon = If(AboveCP, CouponRateHigh, CouponRateLow)
    FlowValue = Interest(Notional, (PayRec * RealizedCoupon), Curr, 1, 'DGT_COUPON', 1)
    EQLeg += FlowValue
    Option += FlowValue
  EndIf
IR_CouponPeriod:
  If Not(KNOCKED_OUT) Then
    FlowValue = Interest(Notional, (-PayRec * (IR_CouponRate + IR_Spread)), Curr, 1,
 '', 2)
    IRLeg += FlowValue
    Option += FlowValue
  EndIf
KO:
  If Not(KNOCKED_OUT) Then
    If (Index >= KO_Barrier) Then
      KNOCKED_OUT = True
      Prob_KO = 1
      FlowValue = Principal((((PayRec * Notional) * KO_FX) * (PrincipalProtection +
 KO_Rebate)), KO_Curr, 1, 'KNOCK_OUT', 1)
      EQLeg += FlowValue
      Option += FlowValue
      FlowValue = Principal((-PayRec * Notional), Curr, 1, 'REDEMPTION', 2)
      IRLeg += FlowValue
      Option += FlowValue
    EndIf
  EndIf
Maturity:
  If Not(KNOCKED_OUT) Then
    Performance = (Index / Strike)
    If Not(KNOCKED_IN) Then
      FlowValue = Principal(((PayRec * Notional) * ((PrincipalProtection +
 (ITMParticipation * Max((Performance - 1), 0))) + FinalCouponRate)), Curr, 1,
 'REDEMPTION', 1)
      EQLeg += FlowValue
      Option += FlowValue
      FlowValue = Principal((-PayRec * Notional), Curr, 1, 'REDEMPTION', 2)
      IRLeg += FlowValue
      Option += FlowValue
    Else
      Prob_KI = 1.0
      If (Index >= Strike) Then
        FlowValue = Principal(((PayRec * Notional) * PrincipalProtection), Curr, 1,
 'REDEMPTION', 1)
        EQLeg += FlowValue
        Option += FlowValue
        FlowValue = Principal((-PayRec * Notional), Curr, 1, 'REDEMPTION', 2)
        IRLeg += FlowValue
        Option += FlowValue
      Else
        Select Case Settlement
          Case 'Physical'
```

```
            DelivQty = (Notional / Strike)
            FlowValue = Physical((PayRec * DelivQty), Index, 0.0, 1, 'REDEMPTION', 1)
            EQLeg += FlowValue
            Option += FlowValue
          Case 'Cash'
            FlowValue = Principal(((PayRec * Notional) * Performance), Curr, 1,
 'REDEMPTION', 1)
            EQLeg += FlowValue
            Option += FlowValue

         EndSelect
        FlowValue = Principal((-PayRec * Notional), Curr, 1, 'REDEMPTION', 2)
        IRLeg += FlowValue
        Option += FlowValue
      EndIf
    EndIf
  EndIf
```

Script: BOEvents

KNOCK_OUT|KNOCKED_OUT
KNOCK_IN|KNOCKED_IN

Script: BarrierDescriptors

KO|KO_Barrier|Index|Up|Out|Closing|
KI|KI_Barrier|Index|Down|In|Closing|